unittest
{
	assert( [ "2", "1 5", "2 4", "3 6" ].parse.expand.solve == 3 );
	assert( [ "3", "1 1 1", "2 2 2", "3 3 3" ].parse.expand.solve == 27 );
	assert( [ "6", "3 14 159 2 6 53", "58 9 79 323 84 6", "2643 383 2 79 50 288" ].parse.expand.solve == 87 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.string;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.strip.to!( long );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	input.popFront;
	auto bs = input.front.split.to!( long[] );
	input.popFront;
	auto cs = input.front.split.to!( long[] );
	return tuple( as, bs, cs );
}

auto solve( long[] as, long[] bs, long[] cs )
{
	auto r = 0L;
	auto ai = 0L;
	auto ci = 0L;
	
	as.sort();
	bs.sort();
	cs.sort();
	
	// Bを基準にする
	foreach( b; bs )
	{
		// Bより小さいAの数を検索
		while( ai < as.length && as[ ai ] < b ) ai++;
		
		// Bより大きいCの数を検索
		while( ci < cs.length && cs[ ci ] <= b ) ci++;
		auto cr = cs.length - ci;
		
		// 組み合わせの数を計算
		r += ai * cr;
	}
	
	return r;
}
